The Universal PDDL Domain

Patrik Haslum Australian National University firstname.lastname@anu.edu.au

Augusto B. Corrêa University of Basel, Switzerland augusto.blaascorrea@unibas.ch

Abstract

In AI planning, it is common to distinguish between planning domains and problem instances, where a "domain" is generally understood as a set of related problem instances. This distinction is important, for example, in generalised planning, which aims to find a single, general plan or policy that solves all instances of a given domain. In PDDL, domains and problem instances are clearly separated: the domain defines the types, predicate symbols, and action schemata, while the problem instance specifies the concrete set of (typed) objects, the initial state, and the goal condition. In this paper, we show that it is quite easy to define a PDDL domain such that any propositional planning problem instance, from any domain, becomes an instance of this (lifted) "universal" domain. We construct different formulations of the universal domain, and discuss their implications for the complexity of lifted domain-dependent or generalised planning.

1. Introduction

In AI planning, a distinction is often made between planning domains and problem instances, where a "domain" is intuitively understood to be a set, typically infinite, of related or similar problem instances. This concept is important in, for instance, planning with domain-specific control knowl-edge (Bacchus & Kabanza, 1995; Doherty & Kvanström, 1999; Wilkins & desJardins, 2000), and in generalised planning, which seeks a single, general plan or policy that solves all instances of a given domain (Srivastava, Immerman, & Zilberstein, 2011). It is materialised in many modelling languages for specifying planning problems, such as PDDL (Haslum, Lipovetzky, Magazzeni, & Muise, 2019), in which the domain and problem instance are syntactically separate. In PDDL, the domain definition contains types, and parameterised predicates and action schemata. The problem instance definition provides the concrete set of (typed) objects, the initial state and the goal condition.

Grundke, Röger, and Helmert (2024) and Haslum and Scholz (2003) argue that PDDL's notion of domain is too weak, in that it does not always allow the modeller to explicitly state the constraints necessary to define precisely the intended set of problem instances, such as constraints on intended "valid" initial states and goals.

Here, we will show that PDDL's notion of domain is also in another sense too general: specifically, that it is possible (indeed, quite trivial) to define a domain such that *any* planning problem instance, of any domain, is an instance of this "universal" domain. There is, however, is caveat: While the universal domain is a parameterised PDDL domain, consisting of types, predicates and action schemata, instances of this domain are arbitrary propositional planning problems. This means that although any PDDL domain–problem pair can be turned into an instance of the universal domain,

```
(define (domain planning)
  (:types action proposition)
  (:predicates (pre ?a - action ?p - proposition)
               (add ?a - action ?p - proposition)
               (del ?a - action ?p - proposition)
               (true ?p - proposition))
  (:action apply
   :parameters (?a - action)
   :precondition (forall (?p - proposition)
                          (imply (pre ?a ?p) (true ?p)))
   :effect (and (forall (?p - proposition))
                         (when (add ?a ?p)
                           (true ?p)))
                (forall (?p - proposition)
                         (when (and (del ?a ?p) (not (add ?a ?p)))
                           (not (true ?p)))))
  )
 )
```

Figure 1: The universal domain for propositional planning.

doing so requires grounding it, with the consequent potentially exponential increase in size. We will also argue that it is not possible to define a universal domain in PDDL such that any domain– problem pair can be expressed as an instance of this domain of a size that is polynomial in that of the domain–problem pair.

2. The Universal Propositional Planning Domain

The universal PDDL domain for propositional planning, in its simplest form, is shown in Figure 1. It has two types: action and proposition, representing the (ground) actions and propositions of the problem instance, respectively. It has a single action schema, apply, with one parameter ?a of type action. An instance of this action schema with ?a = a is applicable iff every proposition of the problem instance is either true in the current state or not a precondition of the action a, i.e., iff a is applicable. Its effect is to make true every proposition that is an add effect of a, and false every proposition that is a delete effect, and not an add effect, of a (implementing PDDL's delete-before-add semantics). Thus, the effect of (apply a) is exactly the effect of a.

Given a propositional planning problem instance Π , an instance P_{Π} of the universal domain is constructed with all ground actions and propositions in Π as objects, initial state facts (pre a p), (add a p) and (del a p) for all ground actions a and propositions p such that $p \in \text{pre}(a), p \in \text{add}(a)$ and $p \in \text{del}(a)$, respectively, and (true p) for each proposition true in the initial state of Π , and goal (and (true p_1)...(true p_m), where p_1, \ldots, p_m are the goal facts of Π . It is easy to see that this instance has a plan iff Π has a plan, and the plan for Π is in fact simply the sequence of arguments of the actions in the plan for P_{Π} .

```
(define (problem sussman)
 (:domain planning)
 (:objects ontable A ontable B ontable C on A B on A C
           on_B_A on_B_C on_C_A on_C_B clear_A clear_B clear_C
            holding_A holding_B holding_C hand_empty - proposition
           pickup_A pickup_B pickup_C putdown_A putdown_B putdown_C
stack_A_B stack_A_C stack_B_A stack_B_C stack_C_A stack_C_B
            unstack_A_B unstack_A_C unstack_B_A unstack_B_C unstack_C_A
            unstack C B - action)
 (:init
  (pre pickup A ontable A) (pre pickup A clear A) (pre pickup A hand emtpy)
                                                                                (add pickup A holding A)
   (del pickup_A ontable_A)
                            (del pickup_A clear_A)
                                                     (del pickup_A hand_empty)
                            (pre pickup_B clear_B)
                                                     (pre pickup_B hand_emtpy)
   (pre pickup_B ontable_B)
                                                                                 (add pickup_B holding_B)
                            (del pickup B clear B)
                                                     (del pickup B hand empty)
   (del pickup B ontable B)
                             (pre pickup_C clear_C)
   (pre pickup_C ontable_C)
                                                     (pre pickup_C hand_emtpy)
                                                                                (add pickup_C holding_C)
   (del pickup C ontable C) (del pickup C clear C)
                                                     (del pickup C hand empty
   (pre putdown_A holding_A)
                             (add putdown_A ontable_A) (add putown_A clear_A)
                              (del putdown_A holding_A)
   (add putdown_A hand_empty)
   (pre putdown B holding B)
                             (add putdown_B ontable_B) (add putown_B clear_B)
   (add putdown_B hand_empty)
                              (del putdown_B holding_B)
   (pre putdown_C holding_C) (add putdown_C ontable_C) (add putown_C clear_C)
   (add putdown C hand empty) (del putdown C holding C)
   (pre stack_A_B holding_A)
                             (pre stack_A_B clear_B)
                                                       (add stack_A_B on_A_B) (add stack_A_B clear_A)
   (add stack_A_B hand_empty)
                              (del stack_A_B holding_A) (del stack_A_B clear_B)
   (pre stack_A_C holding_A)
                             (pre stack A C clear C) (add stack A C on A C) (add stack A C clear A
   (add stack_A_C hand_empty)
                               (del stack_A_C holding_A) (del stack_A_C clear_C)
   (pre stack_B_A holding_B) (pre stack_B_A clear_A) (add stack_B_A on_B_A) (add stack_B_A clear_B)
   (add stack_B_A hand_empty) (del stack_B_A holding_B) (del stack_B_A clear_A)
   (pre stack_B_C holding_B)
                              (pre stack_B_C clear_C)
                                                      (add stack_B_C on_B_C) (add stack_B_C clear_B)
   (add stack B C hand empty)
                              (del stack_B_C holding_B) (del stack_B_C clear_C)
   (pre stack_C_A holding_C) (pre stack_C_A clear_A) (add stack_C_A on_C_A) (add stack_C_A clear_C)
   (add stack_C_A hand_empty)
                              (del stack_C_A holding_C) (del stack_C_A clear_A)
   (pre stack C B holding C) (pre stack C B clear B) (add stack C B on C B) (add stack C B clear C
   (add stack_C_B hand_empty) (del stack_C_B holding_C) (del stack_C_B clear_B)
   (pre unstack_A_B on A B)
                            (pre unstack_A_B clear_A) (pre unstack_A_B hand_empty) (add unstack_A_B holding_A)
   (add unstack_A_B clear_B) (del unstack_A_B on_A_B) (del unstack_A_B clear_A)
   (pre unstack_A_C on_A_C) (pre unstack_A_C clear_A)
                                                        (pre unstack_A_C hand_empty) (add unstack_A_C holding_A)
   (add unstack A C clear C) (del unstack A C on A C) (del unstack A C clear A)
                                                        (pre unstack_B_A hand_empty) (add unstack_B_A holding_B)
   (pre unstack_B_A on_B_A) (pre unstack_B_A clear_B)
   (add unstack B A clear A)
                             (del unstack_B_A on_B_A)
                                                        (del unstack_B_A clear_B)
   (\texttt{pre unstack}\_B\_C \texttt{ on}\_B\_C) \quad (\texttt{pre unstack}\_B\_C \texttt{ clear}\_B)
                                                        (pre unstack_B_C hand_empty) (add unstack_B_C holding_B)
   (add unstack_B_C clear_C) (del unstack_B_C on_B_C)
                                                        (del unstack_B_C clear_B)
   (pre unstack_C_A on_C_A) (pre unstack_C_A clear_C)
                                                        (pre unstack_C_A hand_empty)
                                                                                       (add unstack_C_A holding_C)
   (add unstack C A clear A) (del unstack C A on C A)
                                                        (del unstack C A clear C)
                                                        (pre unstack_C_B hand_empty)
                                                                                       (add unstack_C_B holding_C)
   (pre unstack_C_B on_C_B) (pre unstack_C_B clear_C)
   (add unstack_C_B clear_B) (del unstack_C_B on_C_B)
                                                        (del unstack_C_B clear_C)
   (true ontable A) (true on C A) (true clear C) (true ontable B) (true clear B))
 (:goal (and (true on_A_B) (true on_B_C)))
```

Figure 2: Example of a problem instance of the universal propositional planning domain.

An example of an instance of the universal domain, representing a classical planning problem, is shown in Figure 2.

A STRIPS formulation of the universal propositional planning domain, without quantified or conditional effects, can be obtained using the idea from Nebel's (2000) polynomial size compilation, but simplified because all predicates in effect conditions are static. A possible formulation is shown in Figure 3. This formulation assumes each action has at least one add effect. In this formulation, application of a ground action a is done by first sequentially checking each of its precondition propositions hold in the current state, then sequentially applying its delete and add effects. The predicate idle represents that no action application is in progress, and must be true in the initial state and goal of an instance of this domain.

Alternatively, we can construct a *parameterised* universal PDDL domain, $D_{p,a,d}$, with one single action schema, two predicate symbols, and no types, using only the STRIPS subset of PDDL. Instances of this domain can encode all propositional planning tasks with at most p propositions in any actions' precondition, a propositions in any add effect, and d propositions in any delete effect.

```
(define (domain planning)
  (has-no-pre ?a)
                       (next-add ?a - action ?p - proposition)
(next-add ?a - action ?p - proposition ?q - proposition)
(last-add ?a - action ?p - proposition)
                       (first-del ?a - action ?p - proposition)
(next-del ?a - action ?p - proposition ?q - proposition)
                       (last-del ?a - action ?p - proposition)
                       (has-no-del ?a)
                       (true ?p - proposition)
                            control predicates
                       (idle)
                       (check-pre ?a - action ?p - proposition)
(apply-add ?a - action ?p - proposition)
                       (apply-del ?a - action ?p - proposition)
   (:action check-first-pre
    :parameters (?a - action ?p - proposition)
    :precondition (and (idle) (first-pre ?a ?p) (true ?p))
:effect (and (not (idle)) (check-pre ?a ?p)))
   (:action check-next-pre
    :parameters (?a - action ?p - proposition ?q - proposition)
    :precondition (and (check-pre ?a ?p) (next-pre ?a ?p ?q) (true ?q))
:effect (and (not (check-pre ?a ?p)) (check-pre ?a ?q)))
   (:action skip-check-pre
    :parameters (?a - action ?p - proposition)
    :precondition (and (idle) (has-no-pre ?a) (first-del ?a ?p))
    :effect (and (not (idle)) (apply-del ?a ?p)))
   (:action apply-first-del
    :parameters (?a - action ?p - proposition ?q - proposition)
:precondition (and (check-pre ?a ?p) (last-pre ?a ?p) (first-del ?a ?q))
:effect (and (not (check-pre ?a ?p)) (apply-del ?a ?q) (not (true ?q))))
   (:action apply-next-del
    :parameters (?a - action ?p - proposition ?q - proposition)
:precondition (and (apply-del ?a ?p) (next-del ?a ?p ?q))
:effect (and (not (apply-del ?a ?p)) (apply-del ?a ?q) (not (true ?q))))
   (:action skip-apply-del
    (action skip appry def
:parameters (?a - action ?p - proposition ?q - proposition)
:precondition (and (check-pre ?a ?p) (last-pre ?a ?p) (has-no-del ?a) (first-add ?a ?p))
:effect (and (not (check-pre ?a ?p)) (apply-add ?a ?q)))
   (:action skip-check-pre-and-apply-del
    :prameters (?a - action ?p - proposition)
:precondition (and (idle) (has-no-pre ?a) (has-no-del ?a) (first-add ?a ?p))
:effect (and (not (idle)) (apply-add ?a ?q)))
   (:action apply-first-add
    :parameters (?a - action ?p - proposition ?q - proposition)
:precondition (and (apply-del ?a ?p) (last-del ?a ?p) (first-add ?a ?q))
:effect (and (not (apply-del ?a ?p)) (apply-add ?a ?q) (true ?q)))
   (:action apply-next-add
    :parameters (?a - action ?p - proposition ?q - proposition)
:precondition (and (apply-add ?a ?p) (next-add ?a ?p ?q))
:effect (and (not (apply-add ?a ?p)) (apply-add ?a ?q) (true ?q)))
   (:action finish
    :parameters (?a - action ?p - proposition)
:precondition (and (apply-add ?a ?p) (last-add ?a ?p))
:effect (and (not (apply-add ?a ?p)) (idle)))
  )
```

Figure 3: A STRIPS formulation of the universal propositional planning domain.

```
(define (domain parameterised-strips-planning-3-2-1)
 (:predicates (ground-action ?prel ?pre2 ?pre3 ?add1 ?add2 ?del1)
               (true ?p))
 (:action apply
  :parameters (?prel ?pre2 ?pre3 ?add1 ?add2 ?del1)
              :precondition (and (ground-action ?pre1 ?pre2 ?pre3 ?add1 ?add2 ?del1)
                    (true ?pre1) (true ?pre2) (true ?pre3))
  :effect (and (true ?add1) (true ?add2) (not (true ?del1)))
    )
)
```

Figure 4: A STRIPS formulation of the parameterised universal planning domain $D_{3,2,1}$.

Figure 4 shows the domain $D_{3,2,1}$ (i.e., parameterised universal domain with p = 3, a = 2, d = 1). In an instance $P_{\Pi}^{D_{3,2,1}}$ of this universal domain, the set of objects corresponds to the propositions of Π . Each initial state atom with predicate symbol ground-action exactly describes a ground action of Π . Moreover, the single action schema apply checks that the preconditions of the ground action are true in the current state, and applies the effects according to the ground action.

Note that our example $D_{3,2,1}$ is already sufficient to represent common encodings of Turing Machines as propositional planning problems (Bylander, 1994). In fact, even $D_{2,1,1}$ is sufficient. Therefore, we can immediately conclude a few complexity results for lifted planning, for example, that it is enough to have one single action schema and one fluent (i.e., non-static) predicate to reach PSPACE-hardness.

3. The Impossibility of a Lifted Universal PDDL Domain

One of the few restrictions that a PDDL domain does impose on all instances of the domain is a fixed maximum arity of predicates. This implies a limit on the length of the shortest plans required to solve any instance of the domain. Let D be a PDDL domain description, and suppose the maximum arity of any predicate in D is k. Let P be a PDDL problem description, that is an instance of D, and m the number of objects in P. Note that $k \leq |D|$ and $m \leq |P|$. We know that the grounding of (D, P) can have at most m^k propositions, and therefore the length of a shortest plan for (D, P) is bounded by $2^{(m^k)}$.

However, we also know that it is possible to construct a PDDL domain D' and problem P' requiring a shortest plan of length $2^{2^{(n-1)}}$, where $|D| + |P| \le c(n \log n)$, for some constant c. (This construction was first demonstrated by Erol, Nau, and Subrahmanian (1991), and can also be found in Section 2.5.4 of the book by Haslum et al. (2019).)

Suppose there exists a universal domain for lifted PDDL planning: the maximum arity of any predicate in this domain is a fixed constant k. Therefore, an instance $P_{(D',P')}$ of this domain representing the domain and problem (D', P') must have a number of objects m such that $m^k \ge 2^{n-1}$, implying $m \ge 2^{(n-1)/k}$, and hence that the size of $P_{(D',P')}$ must grow exponentially in the size of D' and P'.

Note, however, that a more expressive formalism than classical PDDL, such as, for example, a language with recursive terms (function symbols), may not have a corresponding bound on plan

length, and therefore may well have the capacity to formulate a universal domain that admits any domain–problem pair as a polynomial-size instance.

4. Discussion

What are the implications of the existence of the universal planning domain?

First, it demonstrates that domain-dependent planning, or generalised planning, is, in the general case, and with PDDL's definition of domain, futile: there exists domains for which there is no domain-specific strategy or solution algorithm more efficient than one that works for every domain, and for which the only generalised plan that exists is a domain-independent planner.

We can of course classify domains by the computational complexity (hardness) of generalised or domain-specific planning for them, i.e., of the set of instances that they admit. Such studies have been made on a range of commonly used planning benchmark domains (Helmert, 2001, 2006). A very small number of works have identified fragments of lifted PDDL planning that are tractable. For example, Lauer, Torralba, Fišer, Höller, Wichlacz, and Hoffmann (2021) show that delete-free planning in a domain with at-most-unary predicates is tractable. Our universal domain constructions complement these results, by showing that domains remain PSPACE-hard also under a variety of syntactic restrictions. Tractability of a restricted class of domains, however, is typically obtained from the existence of a general, i.e., domain-independent, polynomial-time algorithm that solves instances of domains satisfying the restriction.

The impact on the complexity of domain-specific lifted or generalised planning of restrictions in between these two, somewhat extreme, cases – analysing specific domains vs. coarse syntactic restrictions on PDDL domain formulations – is, as far as we know, mostly unexplored. Jonsson and Bäckström (1996) make an important observation: They study the existence of a "universal plan" (essentially, policy) for propositional planning problems, and show that a universal plan that is both compact (polynomial-size) and efficient (evaluable in polynomial time) does not exist for arbitrary propositional planning problems, but can be constructed "for planning problems such that [optimal-length plan generation] can be solved in polynomial time". If we understand "planning problems" here as the family of instances of a given (PDDL) domain, this suggests that generalised planning is possible for, and only for, domains that encode tractable underlying problems. Characterising the class of such domains by syntactic restrictions is likely to be challenging (though perhaps possible via descriptive complexity theory).

References

- Bacchus, F., & Kabanza, F. (1995). Using temporal logic to control search in a forward chaining planner. In Proc. 3rd European Workshop on Planning (EWSP'95).
- Bylander, T. (1994). The computational complexity of propositional STRIPS planning. *Artificial Intelligence*, 69(1–2), 165–204.
- Doherty, P., & Kvanström, J. (1999). TALplanner: An empirical investigation of a temporal logicbased forward chaining planner. In 6th International Workshop on Temporal Representation and Reasoning, pp. 47–54.
- Erol, K., Nau, D., & Subrahmanian, V. (1991). Complexity, decidability and undecidability results for domain-independent planning: A detailed analysis. Tech. rep. CS-TR-2797, Computer

Science Department, University of Maryland.

- Grundke, C., Röger, G., & Helmert, M. (2024). Formal representations of classical planning domains. In *To appear in Proc. of the 34th International Conference on Automated Planning and Scheduling (ICAPS).*
- Haslum, P., Lipovetzky, N., Magazzeni, D., & Muise, C. (2019). An Introduction to the Planning Domain Definition Language. Morgan & Claypool.
- Haslum, P., & Scholz, U. (2003). Domain knowledge in planning: Representation and use. In *Proc. ICAPS 2003 Workshop on PDDL*.
- Helmert, M. (2001). On the complexity of planning in transportation domains. In *Proc. of the 6th European Conference on Planning (ECP'01)*.
- Helmert, M. (2006). New complexity results for classical planning benchmarks. In *Proc. 16th International Conference on Automated Planning and Scheduling (ICAPS)*, pp. 52–61.
- Jonsson, P., & Bäckström, C. (1996). On the size of reactive plans. In *Proc. AAAI Conference*, pp. 1182–1187.
- Lauer, P., Torralba, A., Fišer, D., Höller, D., Wichlacz, J., & Hoffmann, J. (2021). Polynomial-time in PDDL input size: Making the delete relaxation feasible for lifted planning. In Proc. 30th International Joint Conference on Artificial Intelligence (IJCAI), pp. 4119–4126.
- Nebel, B. (2000). On the compilability and expressive power of propositional planning formalisms.. *Journal of AI Research*, *12*, 271–315.
- Srivastava, S., Immerman, N., & Zilberstein, S. (2011). A new representation and associated algorithms for generalized planning. *Artificial Intelligence*, 175(2), 393–401.
- Wilkins, D., & desJardins, M. (2000). A call for knowledge-based planning. In Proc. AIPS Workshop on Analysing and Exploiting Domain Knowledge for Efficient Planning.